

Transcrivez une partition musicale en programme Oric

avec un exemple : "C'est la mère Michel"

par André C.

Le CEO-mag et la musique

Le mag ne manque pas d'articles consacrés à la musique sur Oric. La plupart de ces articles ont été écrits par des musiciens chevronnés et sont souvent difficiles à comprendre.

Comme moi, beaucoup d'Oriciens ne sont au top en matière de musique. Ayant un peu expérimenté ces derniers temps, je me suis dit que mes explications seraient peut-être plus accessibles au plus grand nombre que celles des spécialistes.

Toutefois, je n'ai rien inventé, je me suis appuyé sur deux excellents articles de Dominique Pessan :

- "PLAY, MUSIC, SOUND" (CEO-mag n°119 d'avril 2000, pages 35 à 39, qui corrige et précise la syntaxe de ces trois commandes (les manuels sont truffés d'erreurs).
- "Interruptions musicales" (CEO-mag n°99 de juillet 1998, pages 20 à 22) qui explique clairement comment s'y prendre pour jouer de la musique en faisant autre chose. Et par la même occasion cet article montre comment faire de la musique à partir d'un programme en langage machine.

Je vous recommande chaudement de lire ou relire ces deux articles.

Les partitions sur Internet

Pour ne pas accumuler les difficultés, je ne me suis pas lancé dans la composition et j'ai préféré

programmer un air connu. Mon choix s'est porté sur "C'est la mère Michel", dont la partition est disponible sur Internet.

Mais attention, de nombreuses partitions sont pleines d'erreurs. Certes, je sais qu'il existe des variantes pour tous les airs traditionnels, mais là, je parle de fausses notes (à l'oreille ça ne va vraiment pas). Lorsqu'on n'est pas musicien, il est impossible de détecter ces fausses notes en lisant une partition.

Il vaut donc mieux se limiter aux sites sérieux.

Wikipédia, par exemple, propose nombre de pages consacrées aux airs traditionnels, avec paroles, musique, historique, etc. Voyez par exemple : https://fr.wikipedia.org/wiki/Catégorie:Chanson_enfantine_en_français (58 chansons).

https://fr.wikipedia.org/wiki/Catégorie:Chanson_populaire_française (91 chansons).

Quelques notions élémentaires de musique

Très brièvement, une note de musique est caractérisée par sa hauteur (do, ré, mi, etc.), sa durée (ronde, blanche, noire, croche, double-croche, etc.), son intensité (le volume sonore) et son timbre (c'est ce qui différencie l'accordéon, de la flûte, etc.).

1) *Hauteur* : Nous nous en tiendrons à la clé de sol, la plus courante (Figure 1) :




Figure 1

En fait, dans une octave, il n'y a pas 8 (octo) mais 12 notes numérotées de 1 à 12 :

do, do#, ré, ré#, mi, fa, fa#, sol, sol#, la, la# et si (voir le manuel Atmos, page 115).

Sur la partition, les # sont notés soit immédiatement à gauche d'un note et concerne toutes les notes de même nom et de même hauteur situées dans la même mesure, soit immédiatement après la clé et dans ce dernier cas ils

concernent toutes les notes de la ligne.

Les histoires de dièse, bémol et bécarre sont en fait très compliquées et pas pour nous aujourd'hui !

En théorie l'Oric peut couvrir 7 octaves (pas tous bien beaux hélas). En pratique, pour la clé de sol, les octaves les plus courantes sont les n°3 et 4 (Voir la figure 2 page suivante).

Les notes de 1 à 12 pour les octaves n°3 et n°4

Figure 2

La figure figure 2 ci-dessus indique, en dessous de chaque note, le n° d'octave (3 ou 4) et le numéro de la note (1 à 12) dans cette octave.

2) *Durée* : La Figure 3 montre les notes et les silences que l'on trouve sur une partition : Comme vous pouvez le voir, la durée des notes est relative. La ronde sert de référence et sa durée est marqué "1". La durée des autres notes est un multiple ou un sous-multiple de cette durée de référence.

Une ronde vaut deux blanches, une blanche vaut deux noires, etc. Il en est de même pour les silences, la pause ayant la même durée qu'un ronde, etc.

Certaines notes sont suivies d'un point. Ce point signifie que la durée de cette note doit être augmentée de 50 %. La durée totale des notes situées entre 2 barres verticales (une mesure) est constante pour une partition donnée.

Notez bien qu'aucune valeur absolue n'est donnée pour la durée de notes et des silences, car la durée dépend du tempo (vitesse d'exécution) en général noté sur la partition (adagio, allegro, etc.). Chacune de ces indications correspond à une fourchette de vitesse d'exécution. Par exemple allegro signifie que l'on doit jouer de 112 à 160 croches / minute.

L'indication du tempo est parfois même notée de manière très précise. Par exemple ♩ = 120 signifie que l'on doit jouer en une minute l'équivalent de cent-vingt croches (0,5s chaque) : Dans cet exemple, une noire dure 1s, une blanche dure 2s, une ronde dure 4s, etc.

Mais pour ce qui nous occupe aujourd'hui, tout cela ne vous est pas bien utile, car le tempo est rarement indiqué sur les partitions des musiques traditionnelles.

3) *Intensité* : Les Oriciens parlent de "volume" (voir la syntaxe des commandes MUSIC et SOUND) et ça va de 1 (faible) à 15 (très fort). La valeur 0 ne coupe pas le son, mais passe le relai

Durée	Note	Figure 3	Silence
2	carrée		bâton de pause
1	ronde		pause
1/2	blanche		demi-pause
1/4	noire		soupir
1/8	croche		demi-soupir
1/16	double-croche		quart de soupir
1/32	triple-croche		huitième de soupir
1/64	quadruple-croche		seizième de soupir
1/128	quintuple-croche		trente-deuxième de soupir

aux paramètres "enveloppe" et "durée" de la commande PLAY.

Dans le cadre de l'exercice d'aujourd'hui, on gardera le même volume pour toutes les notes. Mais une légère modification du programme permettrait facilement de jouer certains passages pianissimo ou fortissimo.

4) *Timbre* : On ne va pas entrer ici dans la simulation de divers instruments par l'Oric. C'est possible, mais pour l'instant on va faire simple.

En pratique, transcription d'une partition vers un programme Basic

C'est pas tout d'avoir une bonne partition, encore faut-il pouvoir la traduire en Oric ! En vous basant d'une part sur les explications précédentes, notamment les figures 2 et 3 et d'autre part en comparant la partition et le programme proposés ci-après, vous pourrez comprendre ça s'articule.

Pour pouvoir vous y retrouver facilement, réservez une ligne de DATA pour chaque mesure. N'hésitez pas à aligner les valeurs avec des espaces, même si cela donne un programme un peu plus gros.

Dans cet exemple, la somme des durées des notes et des silences dans chaque mesure est de 80. C'est un bon moyen de vérifier qu'il n'y a pas d'erreur. Il n'y a pas de silence dans cette partition, mais lorsque vous en rencontrerez un, il sera aisé d'en calculer la durée de manière que la somme des durées des notes et des silences soit constante.

Entrons un peu dans le détail

1) *Hauteur des notes* : La figure 2 devrait vous aider à déchiffrer la partition. En extrapolant vous pourrez facilement identifier celles qui ne sont pas sur la figure, soit plus graves (octave n°2), soit plus aiguës (octave n°4).

Je rappelle que chaque note est caractérisée par deux valeurs : n° d'octave (0 à 7) et n° de note dans l'octave (1 à 12). Il n'y a pas de souci particulier, il faut seulement faire très attention, notamment aux notes altérées (dièses, etc.).

2) *Vitesse d'exécution / durée des notes* : Là, c'est moins simple, car comme déjà indiqué le tempo n'est pas toujours indiqué sur les partitions.

Le manuel Atmos n'est pas bien clair au sujet de la durée des notes. Ce paramètre est pourtant fondamental.

Deux possibilités s'offrent à nous pour contrôler la durée d'une note :

- Soit la commande PLAY avec les enveloppes 1 ou 2 (sinon le son est continu avec les autres enveloppes), mais les résultats ne correspondent pas toujours à ce que l'on voudrait avoir (il faut beaucoup d'expérience).
- Soit la commande WAIT. En effet le son émis par MUSIC ou SOUND dure infiniment. Pour contrôler sa durée il faut utiliser la commande WAIT, suivie de la commande PLAY 0,0,0,0 qui coupe le son.

C'est cette deuxième méthode que je vous conseille d'utiliser tant que vous ne serez pas un as de la programmation musicale.

Oui, mais quelle valeur donner à WAIT ?

En comparant la durée totale d'exécution du programme "C'est la Mère Michel", que j'ai écrit en Basic, à celle de plusieurs enregistrements disponibles sur Internet, j'ai trouvé qu'il faut partir sur la base d'un WAIT 160 pour la durée d'une ronde. Je reconnais que cette méthode d'estimation est évidemment très contestable.

Le manuel indique que l'unité utilisée par WAIT est 1/100e de seconde. J'ai vérifié que WAIT

6000 dure 61s (au lieu de 60 comme attendu, mais c'est dans la limite d'imprécision de ma mesure).

Donc WAIT 160 correspondrait à une durée de 1,6s pour une ronde (la durée des autres notes en découle, voir plus loin).

Mais bien sûr cela reste ENVIRON. En effet les musiciens utilisent un métronome et l'ajuste en fonction des indications portées sur la partition mais aussi de l'idée qu'ils se font du morceau de musique. Donc, essayez de trouver ce qui sonne bien à votre oreille.

Dans notre cas et en simplifiant pour ne garder que le principe, un PLAY x,0,0,0 valide le ou les canaux qui doivent émettre du son, il est suivi d'une ou plusieurs commandes MUSIC qui émet la ou les notes appropriées, puis d'un WAIT qui détermine la durée et cela se termine par un PLAY 0,0,0,0 qui coupe le son.

Toujours dans notre cas, je vous indique la durée des notes dans le tableau ci-dessous (figure 4). Vous pouvez voir sur la partition de "C'est la Mère Michel" (Figure 5), que certaines notes sont suivies d'un point. Comme déjà indiqué plus haut, ce point signifie que la durée de cette note doit être augmentée de 50 %.

Idem, la durée totale des notes et des silences situés entre 2 barres verticales (une mesure) est constante pour une partition donnée. Vous pouvez donc vérifier, mesure par mesure, qu'il n'y a pas d'erreurs.

Les silences sont également gérés par la commande WAIT (en fait PLAY 0,0,0,0 suivi d'un WAIT). Il est donc très simple, en totalisant la durée des notes d'une mesure, de déterminer celle d'un éventuel silence. La figure 4 montre les valeurs dont vous aurez besoin avec WAIT pour obtenir les bonnes durées.

En principe, les notes successives doivent être jouées individuellement, c'est à dire avec une micro-pause entre elles. Mais certaines notes sont reliées entre elles par un trait. Cela indique que ces notes doivent être enchaînées. Dans le programme Basic indiqué plus loin, la micro-pause est obtenue avec un WAIT2 et l'enchaînement avec un WAIT0.

Figure 4

Silences	Notes	Durées	Pointées
Pause	Ronde	160	240
demi-pause	Blanche	80	120
Soupir	Noire	40	60
Demi-soupir	Croche	20	30
Quart de soupir	Double-croche	10	etc.

La partition de "C'est la Mère Michel"

Figure 5



C'est la mèr' Michel qui a per-du son chat, qui crie par la fenêtr' à qui le lui ren-dra. C'est le Père', Lus-tu-cru qui lui a ré-pon-du: Al-lez la mèr' Mi-chel vot' chat n'est pas per-du! Sur l'air du tra la la la, sur l'air du tra la la la, sur l'air du tra déridé-ra et tra la la.

Le programme correspondant

```

1000 ' C'est la mere Michel
1010 V=10:' Volume, Octave, Note, Duree, Silence
1020 FOR I=1 TO 73
1030 PLAY 1,0,0,0
1040 READ O,N,D,S
1050 MUSIC 1,O,N,V:WAIT D:PLAY 0,0,0,0:WAIT S
1060 NEXT
1070 'END
2010 DATA 3,10,30,0, 3,10,10,2, 3,10,20,0, 3,10,20,2: 'C'est la mer' Mi
2020 DATA 3,10,20,0, 3, 5,20,2, 3,10,20,0, 3,12,20,2: 'chel qui a per
2030 DATA 4, 2,40,2, 3,12,40,2: 'du son
2040 DATA 3,10,60,2, 3,10,20,2: 'chat Qui
2050 DATA 3, 8,30,0, 3, 6,10,2, 3, 8,20,0, 3,10,20,2: 'cri par la fe
2060 DATA 3,12,20,0, 4, 2,20,2, 3,12,20,0, 3,10,20,2: 'netre A qui le
2070 DATA 3, 8,40,2, 3, 6,40,2: 'lui ren
2080 DATA 3, 5,60,2, 3, 5,20,2: 'dra C'est
2090 DATA 3,12,20,0, 3,12,20,2, 3,12,20,0, 4, 2,20,2: 'le pere Lus tu
2100 DATA 4, 3,60,2, 4, 3,20,2: 'cru qui
2110 DATA 4, 2,20,0, 3,10,20,2, 4, 5,20,0, 4, 2,20,2: 'lui a re pon
2120 DATA 3,12,60,2, 3,12,20,2: 'du Al
2130 DATA 4, 3,20,0, 4, 3,20,2, 4, 3,20,0, 4, 3,20,2: 'lez la mere Mi
2140 DATA 4, 2,20,0, 4, 5,20,2, 4, 2,20,0, 3,10,20,2: 'chel Vot chat n'est
2150 DATA 3,12,40,2, 4, 5,40,2: 'pas per
2160 DATA 3,10,20,2, 3,10,20,2, 3,10,20,0, 3, 8,20,2: 'du Sur l'air du
2170 DATA 3, 6,40,2, 3, 6,30,0, 3, 6,10,2: 'tra la la
2180 DATA 3, 6,20,0, 3, 8,20,2, 3,10,20,0, 3,12,20,2: 'la Sur l'air du
2190 DATA 3, 5,40,2, 3, 5,30,0, 3, 5,10,2: 'tra la la
2200 DATA 3, 5,60,2, 3, 5,20,2: 'la Sur
2210 DATA 3,10,40,2, 4, 2,40,2: 'l'air du
2220 DATA 4, 5,20,0, 4, 3,20,2, 4, 2,20,0, 3,12,20,2: 'tra de ri de
2230 DATA 3,10,20,0, 4, 5,20,2, 4, 5,20,0, 4, 5,20,2: 'ra Et tra la
2240 DATA 3,10,80,2: 'la

```

Vous trouverez ce listing ainsi que la disquette du programme dans le fichier MereMiche.zip qui accompagne ce CEO-mag. Le listing pourra vous servir de base pour adapter une autre chanson.

Et maintenant ?

A titre d'exercice, on pourrait remplacer les commandes MUSIC par des commandes SOUND. Et pendant que nous y sommes, il faudrait améliorer cette musique en ajoutant des harmoniques, ce que SOUND permet. En effet SOUND ne travaille pas

au niveau des notes, mais des fréquences. C'est nettement moins facile, mais cela offre plus de possibilité. Quant à transcrire ce programme en langage machine, cela ne présente aucun intérêt, sauf pour la jouer en bi-tâche, par exemple jeu d'arcade avec une musique de fond. Pour cela voyez l'article de Dominique Pessan, cité en introduction.

Enfin je vous invite à transcrire d'autres airs afin de tester la faisabilité de la méthode et d'enrichir le répertoire musical de l'Oric...

Bon amusement.